

Figure 1: Goodness-of-fit plots for intravenous data for i) high, ii) intermediate 1, iii) intermediate 3, iv) low bioavailable compounds




Figure 2: Goodness-of-fit plots for solution formulation data for i) high, ii) intermediate 1, iii) intermediate 3, iv) low bioavailable compounds




Figure 3A: Goodness-of-fit plots for formulations of intermediate 1; i) IR in the base form; ii) IR in the salt form; iii) IR in the base form at elevated pH, iv) IR in the salt form at elevated pH




Figure 3B: Goodness-of-fit plots for formulations of intermediate 2: i) oral solution fasted state; ii) oral solution fed state; iii) PR in fasted state; iv) PR in fed state




Figure 3C: Goodness-of-fit plots for PR formulation of the low bioavailable compound.

	CLi	V	Q1	V1	Q2	V1
IV high	0.000	0.000	13.291	14.965	16.350	5.521
IV INT1	4.035	0.043	22.044	25.516	10.395	23.487
IV INT2	2.591	0.055	12.206	2.398	10.000	12.000
IV LOW	2.251	2.821	6.210	12.836	25.244	10.000

Table I: ETA shrinkage values for disposition models of the four compounds

Example for run file for disposition model

```
$PROBLEM IV only
$INPUT ID IDORIG TIME AMT RATE WT DV EVID AGE HGT BMI BOQ OCMT
DOSEID FORM
$DATA TESA_IV_PO_dataADME SOLUTION_070214.csv IGNORE=@
$DATA IGNORE=(FORM==2)
$DATA IGNORE=(DV<0)
$SUBROUTINE ADVAN7 TRANS1
$MODEL COMP=(CENTRAL,DEFDOS,DEFOBS) COMP=(PERI1) COMP=(PERI2)
$PK
TVCLI = THETA(1) ; typical value of CLI
TVV = THETA(2)
TVQ1 = THETA(3)
TVV1 = THETA(4)
TVQ2 = THETA(5)
TVV2 = THETA(6)
;
CLI = TVCLI*EXP(ETA(1)) ; individual value of CLI
V = TVV*EXP(ETA(2))
Q1 = TVQ1*EXP(ETA(3))
V1 = TVV1*EXP(ETA(4))
Q2 = TVQ2*EXP(ETA(5))
V2 = TVV2*EXP(ETA(6))
;
LV = 0.05012*WT**0.78      ; Noda liver vol
BPR = 0.7 ; CB/CP=0.7      ; blood to plasma ratio
FQ = 50.4*LV*BPR          ; Price liver blood flow males
CLR = 0.03*(WT/70)**0.75   ; renal cl in L/h normalized to mean weight
;
CLH = FQ*CLI/(FQ+CLI)
ER = CLI/(FQ+CLI)
CL = CLH + CLR

:RATE CONSTANTS
K10 = CL/V
K12 = Q1/V
K21 = Q1/V1
K13 = Q2/V
K31 = Q2/V2

$ERROR
IPRED = A(1)/V
Y = IPRED + IPRED*EPS(1) + EPS(2)
;

$THETA (0,0.195) ; 1. TVCLI (lower bound,initial estimate)
$THETA (0,4.08) ; 2. TVV
$THETA (0,0.439) ; 3. TVQ1
$THETA (0,5.12) ; 4. TVV1
$THETA (0,0.55) ; 5. TVQ2
$THETA (0,1.52) ; 6. TVV2
```

```

;
$OMEGA 0.0447 ; 1. CLI
$OMEGA 0.0464 ; 2. V
$OMEGA 0.0189 ; 3. Q1
$OMEGA 0.00766 ; 4. V1
$OMEGA 0.1 ; 5. Q2
$OMEGA 0.1 ; 6. V2
;
$SIGMA 0.00241 ; variance PROP res error, initial estimate
$SIGMA 10 ; additive residual error
$ESTIMATION METHOD=1 INTER MAXEVAL=9999 PRINT=1 ; calculation method
; standard error of estimate is calculated
$COVARIANCE
$TABLE ID TIME IPRED CWRES EVID NOPRINT ONEHEADER FILE=sdtab1
$TABLE ID CL V Q1 V1 Q2 V2 ETA(1) ETA(2) ETA(3) ETA(4) ETA(5)
ETA(6) NOPRINT NOAPPEND ONEHEADER FILE=patab1
$TABLE ID AGE HGT BMI BOQ OCMT DOSEID FORM NOPRINT NOAPPEND
ONEHEADER FILE=cotab1

```

Example for run file for absorption model

```

$PROBLEM IV & sol
$INPUT ID IDORIG TIME AMT RATE WT DV EVID AGE HGT BMI BOQ CMT
DOSEID FORM
$DATA TESA_IV_PO_dataADME SOLUTION_070214AB.csv IGNORE=@
$DATA IGNORE=(DV<0)
$SUBROUTINE ADVAN5 TRANS1
$MODEL COMP=(DEPOT) COMP=(CENTRAL) COMP=(PERI1) COMP=(PERI2)
$PK
TVCLI = THETA(1) ; typical value of CLI
TVV = THETA(2)
TVQ1 = THETA(3)
TVV1 = THETA(4)
TVQ2 = THETA(5)
TVV2 = THETA(6)
TVKA = THETA(7) ;absorption rate constant
TVFA = THETA(8)
TVALAG1 =THETA(9) ;lag time
;
CLI = TVCLI*EXP(ETA(1)) ; individual value of CLI
V = TVV*EXP(ETA(2))
Q1 = TVQ1*EXP(ETA(3))
V1 = TVV1*EXP(ETA(4))
Q2 = TVQ2*EXP(ETA(5))
V2 = TVV2*EXP(ETA(6))
KA = TVKA*EXP(ETA(7))
FA = 1/(1+EXP(-(TVFA+ETA(8))))
ALAG1 = TVALAG1*EXP(ETA(9))
;
LV = 0.05012*WT**0.78 ; Noda liver vol
BPR = 0.7 ; CB/CP=0.7 ; blood to plasma ratio

```

```

FQ = 50.4*LV*BPR      ; Price liver blood flow males
CLR = 0.03*(WT/70)**0.75 ; renal cl in L/h normalized to mean weight
;
CLH = FQ*CLI/(FQ+CLI)
ER = CLI/(FQ+CLI)
CL = CLH + CLR
F1 = FA*(1-ER)
;
;RATE CONSTANTS
K12 = KA
K20 = CL/V
K23 = Q1/V
K32 = Q1/V1
K24 = Q2/V
K42 = Q2/V2
;
$ERROR
IPRED = A(2)/V
Y    = IPRED + IPRED*EPS(1) + EPS(2)
;
$THETA 0.159677 FIX ; 1. TVCLI
$THETA 4.08887 FIX ; 2. TVV
$THETA 0.435556 FIX ; 3. TVQ1
$THETA 5.09234 FIX ; 4. TVV1
$THETA 0.542979 FIX ; 5. TVQ2
$THETA 1.53715 FIX ; 6. TVV2
$THETA (0,3) ; 7. TVKA
$THETA 1 ; 8. TVFA
$THETA (0,0.111634) ; 9. LAG TIME
;
$OMEGA 0.0687225 FIX ; 1. CLI
$OMEGA 0.0459124 FIX ; 2. V
$OMEGA 0.0179779 FIX ; 3. Q1
$OMEGA 0.00789633 FIX ; 4. V1
$OMEGA 0.0865778 FIX ; 5. Q2
$OMEGA 0.179475 FIX ; 6. V2
$OMEGA 0.1 ; 7. KA
$OMEGA 0.1 ; 8. FA
$OMEGA 0.1 ; 9. ALAG1
;
$SIGMA 0.02; variance PROP res error
$SIGMA 10 ; additive residual error
;
$ESTIMATION METHOD=1 INTER MAXEVAL=9999 PRINT=1
$COVARIANCE MATRIX=R
;
$TABLE ID TIME IPRED CWRES EVID NOPRINT ONEHEADER FILE=sdtab2
$TABLE ID CL V Q1 V1 Q2 V2 KA FA ALAG1 ETA(1) ETA(2) ETA(3)
ETA(4) ETA(5) ETA(6) ETA(7) ETA(8) ETA(9) NOPRINT NOAPPEND
ONEHEADER FILE=patab2
$TABLE ID AGE HGT BMI BOQ CMT DOSEID FORM NOPRINT NOAPPEND
ONEHEADER FILE=cotab2

```

